
Hoofdstuk 1 : 

1.1 Het assenstelsel.

Iedereen is reeds vertrouwd met een assenstelsel uit de lessen wiskunde en fysica. Toch even een korte herhaling.

Om een assenstelsel te tekenen hebben we nodig:

· Een nulpunt, het punt met coördinaten (0,0)

· 2 assen, een x-as, dat is de horizontale as, en een y-as, dat is de verticale as.

· Schaalfactoren voor de beide assen. Dit hebben we nodig om de eenheden op de assen aan te duiden. De schaalfactoren kunnen verschillend zijn voor de x-as en y-as. De schaalfactor komt overeen met het aantal basiseenheden die 1 eenheid op het assenstelsel bevat.

Een voorbeeld is een blad geruit papier waarop we een assenstelsel willen teken. 

Eerst gaan we het nulpunt ergens kiezen op ons blad, voor het gemak nemen we meestal een punt dat op het snijpunt ligt van 2 lijnen. 

Dan gaan we door dat punt een horizontale rechte tekenen, dit wordt de x-as en dan gaan we door het nulpunt een verticale as tekenen, dit wordt de y-as.

Daarna kiezen we voor beide assen een schaalfactor. De basiseenheid op ons blad papier is 1 ruitje. We kiezen bijvoorbeeld als schaalfactor voor de x-as 2 ruitjes en voor de y-as 3 ruitjes. We kunnen nu onze x-as en y-as onderverdelen in stukjes van respectievelijk 2 en 3 ruitjes. We beginnen wel te verdelen vanaf het nulpunt in beide richtingen.

1.2 Het assenstelsel tekenen in LOGO.

Ook in LOGO kunnen we een assenstelsel tekenen. In het LOGO scherm is de basiseenheid dan 1 turtlestapje. Het LOGO scherm bevat eigenlijk al een assenstelsel dat we niet zien, met schaalfactor 1 voor beide assen en met het nulpunt in het midden van het scherm.

We kunnen nu een eigen assenstelsel gaan tekenen met andere schaalfactoren en eventueel een ander nulpunt.

Stel we kiezen als schaalfactoren 20 voor de x-as en 10 voor de y-as. Dwz dat we de x-as gaan verdelen in stukjes van 20 turtlestapjes en de y-as in stukjes van 10 turtlestapjes.

In het oorspronkelijke LOGO scherm kunnen we 500 eenheden aanduiden op de halve as. We kunnen punten aanduiden met coördinaten van –500 tot 500 voor x en y. 

Wanneer we nu zelf een assenstelsel gaan tekenen met als nulpunt het oorspronkelijke nulpunt maar met andere schaalfactoren moeten we ook weten hoeveel eenheden we kunnen aanduiden.

Het aantal eenheden voor een halve as (van het nieuwe assenstelsel) is gelijk aan de halve lengte van de oorspronkelijke as in turtlestapjes gedeeld door de schaalfactor (of het aantal turtlestapjes in 1 eenheid van het nieuwe assenstelsel).

Voor het voorbeeld van daarjuist wordt dat dan 500/20 = 25 eenheden voor de halve x-as en 500/10 = 50 eenheden voor de halve y-as. Met andere woorden we kunnen punten aanduiden met coördinaten van –25 tot 25 voor x en van –50 tot 50 voor y.

Dus willen we nu 5 eenheden aanduiden op de x-as dan moeten we de turtle 20 * 5 stapjes verder zetten.

Als we het punt met coördinaten (2,3) willen aanduiden, dan gaan we 2 * 20 stapjes verder op de x-as en 3 * 10 stapjes op de y-as.

Om nu het nieuwe assenstelsel visueel te maken in LOGO, moeten we de assen tekenen en de eenheden aanduiden. Een logische werkwijze zou zijn : we tekenen eerst de assen en duiden dan de eenheden aan, zoals we doen op een blad papier of op bord. In LOGO kunnen we beter 1 eenheid tekenen in de positieve y-richting ( naar boven ) dan een streepje zetten dan terug 1 eenheid verder gaan en terug een streepje enz. tot we het einde van het scherm bereikt hebben. Dan keren we terug naar het nulpunt en doen het zelfde voor de positieve zijde van de x-as. We herhalen dit alles dan nog een keer voor de negatieve zijden van de assen. We kunnen herhaling gebruiken door de symmetrie die aanwezig is.

1.3 De procedure ASSEN.
Om een assenstelsel te tekenen gaan we een procedure assen maken met als argumenten beide schaalfactoren. ASSEN :XS :YS (xs = schaalfactor voor de x-as en ys= schaalfactor voor de y-as)

De procedure assen ziet er als volgt uit : 


to ASSEN :xs :ys 



make "xschaal :xs



make "yschaal :ys



local "xrep



local "yrep



make "xrep int (xdim / :xschaal)



make "yrep int (ydim / :yschaal)



repeat 2 [yas :yrep rt 90 xas :xrep rt 90]

end
We overlopen deze procedure:

· Eerst slaan we de waarden van de schaalfactoren op in xschaal en yschaal. Zo kunnen we deze waarde ook gebruiken buiten het programma. Eens we de procedure assen 1 keer gebruikt hebben kunnen we de waarden van de schaalfactoren opnieuw oproepen. Met show :xschaal kunnen we dan zien wat de huidige schaalfactor is voor de x-as.

· De twee lokale variabelen (lokaal wil zeggen dat ze enkel binnen deze procedure gebruikt kunnen worden) xrep en yrep bevatten respectievelijk het aantal eenheden dat we kunnen aanduiden op een halve x-as en een halve y-as. Dus het aantal streepjes dat we zullen moeten zetten. Dit is dus het quotiënt van de lengte van de halve as ( dit zit in xdim en ydim) en de schaalfactor.

· In de laatste regel tekenen we dan de assen. Yas tekent een halve y-as , x-as tekent een halve x-as. We tekenen dus eerst de positieve y-as draaien over 90 graden naar rechts en tekenen dan de positieve x-as. We draaien nog eens 90 graden en staan klaar om de negatieve assen te tekenen. Dit is het zelfde als de positieve dus herhalen we gewoon het vorige.

De procedures xdim en ydim zien er als volgt uit : 


to XDIM



output 500

end

to YDIM



output 500

end
Dit zijn twee procedures die gewoon de lengte van de halve as terug geven.

De procedures voor xas en yas zien er als volgt uit :


to XAS :xrep



repeat :xrep [mark :xschaal]



bk product :xrep :xschaal

end

to YAS :yrep



repeat :yrep [mark :yschaal]



bk product :yrep :yschaal

end
Beide procedures zijn analoog.

Wat deze procedures doen is het volgende :

· Ze herhalen xrep keer ( dit is het aantal streepjes op een halve as) mark. De procedure mark zal telkens een platte T tekenen :--| dus zal 1 eenheid tekenen. Na xrep keer mark te doen zal de turtle op het einde van de as staan en we moeten ze dan nog terugzetten naar het nulpunt, we zetten de turtle dus xrep keer xschaal terug.

De procedure mark ziet er zo uit:


to MARK :schaal



fd :schaal cross :schaal

end

Hier gaan we 1 eenheid verder dus de waarde van de schaalfactor bepaalt het aantal stapjes van de turtle en dan zetten we het kruisje of streepje.

De procedure cross doet het volgende:


to CROSS :schaal



local "bar


make "bar :schaal / 10



lt 90


fd :bar



bk 2* :bar



fd :bar



rt 90

end
De lokale variabele bar bevat de grootte van een half streepje, we nemen hier de schaalfactor gedeeld door tien. De gaan we beide helften van het streepje tekenen.

Nu kunnen we het assenstelsel gaan tekenen op ons scherm. We maken eerst ons scherm leeg met cs en dan tekenen we de assen met assen 20 10.

1.4 De Turtle in het assenstelsel bewegen.
Om te weten waar de turtle zich ergens bevindt in het scherm typen we het commando show pos in. Dit geeft dan de coördinaten in het oorspronkelijke assenstelsel terug.

Om de turtle op een bepaalde plaats in het assenstelsel te zetten typen we setpos [x y] in. Met setpos werken we ook in het oorspronkelijke assenstelsel. Wat als we nu de turtle op positie(x,y) willen in het nieuwe assenstelsel ? We moeten dan de coördinaten vermenigvuldigen met de respectievelijke schaalfactoren. De x-coördinaat wordt dan x-coördinaat maal de schaalfactor voor de x-as. Analoog voor de y-as. 


x := x * xschaal


y := y * yschaal

Dus ipv setpos [3 3] moeten we dan setpos [60 30] ingeven. ( als xs = 20 en ys = 10)

We zouden nu elke keer we een positie willen ingeven die berekeningen zelf kunnen maken, beter zou natuurlijk zijn dat we daar een procedure van maken. 

Eerst maken we een procedure die de geschaalde coördinaten teruggeeft als we de schaalfactoren en de coördinaten van het punt meegeven.


to SCHALEN :xs :ys :punt



op list (:xs * first :punt) ( :ys * last :punt)

end
We nemen hier de juiste coördinaat en vermenigvuldigen die met de juiste schaalfactor en geven de resultaten dan terug in een lijst zodat het weer als een punt gezien kan worden.

Deze procedure kunnen we nu gebruiken om de turtle naar de juiste positie te brengen in het nieuwe assenstelsel. We noemen deze procedure sp( setpos). 


to SP :punt



setpos schalen :xschaal :yschaal :punt

end
1.5 Opgave.
1. Teken een assenstelsel met schaalfactoren 20, 20

Opl : assen 20 20

2. Trek nu een lijn van punt(1,1) naar punt (1,5).

Opl : pu
sp[1 1]

pd
sp[1 5] 
pu 
home


3. Teken nu het huisje dat hieronder staat afgebeeld op de juiste plaats.

Oplossing :

to HUISJE


pu
sp [1 1]


pd
sp [11 1]
sp [11 6]
sp [1 6]
sp [1 1]


pu
sp [0 5]


pd
sp [6 11]
sp [12 5]



pu
sp [6 8]
pd

circle :xschaal



pu
sp [2 1] 


pd 
sp [2 4] 
sp [4 4] 
sp [4 1] 


pu
sp [5 2] 


pd 
sp [5 5] 
sp [10 5] 
sp [10 2] 
sp [5 2]


pu
home

pd

end 

HOOFDSTUK 2 :Functies. 

2.1 Definitie

Een functie is iets van de volgende vorm : y = f(x).

Voorbeeld: y = 2x + 5

x en y zijn variabelen.

Voor de functies die we gaan plotten in LOGO gaan we ons beperken tot functies met twee variabelen (x en y) en functies met voor elke x waarde hoogstens één y waarde. 

2.2 Functies tekenen.

Wanneer we een functie willen tekenen in een assenstelsel bepalen we eerst de twee uiterste waarden van x waarvoor we de functie gaan plotten. Anders gezegd we kiezen de minimum en maximum x waarde, de grenzen. We gaan de functie dan plotten vanaf de minimum waarde tot aan de maximum waarde.

Hoe gaan we nu te werk:

· We kiezen een eerste waarde voor x, om het gemakkelijk te maken nemen we de minimum waarde.

· Dan berekenen we f(x) met x de gekozen waarde.

· Als we f(x) kennen duiden we het punt met coördinaten (x, f(x)) aan op het assenstelsel.

· We kiezen nu een volgende x waarde, om het opnieuw gemakkelijk te maken, doen we dit door bij de oude waarde van x een constant getal op te tellen. Vb x :=  x + 1 (met 1 de constante waarde)

· We berekenen de nieuwe f(x) enz….

We gaan zo verder tot de waarde van x het maximum bereikt heeft. Dan gaan we alle aangeduide punten met elkaar verbinden.

Hoe meer punten je aanduidt, dus hoe kleiner je de constante waarde neemt die je telkens optelt, hoe beter het resultaat.

2.3 Functies evalueren.

Een functie evalueren is het berekenen van de waarde van y die overeenkomt met een waarde van x. Maw f(x) berekenen voor een zekere x waarde. We noemen f(x) ook wel de regel.

Om in LOGO nu de waarde van zo een regel te berekenen voor een zekere x waarde gebruiken we het volgende commando : 


Run [regel]

Regel moet dan wel een functie zijn met :x in. Vb : 2 * :X + 5
Om nu de functie y = 2 * x + 5 te evalueren voor de x waarde 7 doen we het volgende:

Eerst maken we de variabele x aan met als waarde 7: make “x 7 

Dan gaan we de regel evalueren met : show run [ 2 * :x + 5] 

Dit geeft dan als resultaat in het tekstscherm : 19

We gaan het evalueren van een functie nu in een procedure steken : 

to EVALX :regel :x


op run :regel

end
We geven dus de regel en de x waarde mee en krijgen als resultaat de y waarde terug.

Als we show evalx [2 * :x + 5] ingeven krijgen we 19 terug.

We kunnen de functie of regel evalueren voor 1 waarde van x , we moeten dit nu nog kunnen aanduiden in het assenstelsel.

Om het punt (x,f(x)) af te beelden moeten we het volgende commando ingeven.


Setpos list :xschaal * :x :yschaal * evalx : regel :x

Om de hele functie af te beelden moeten we dit doen voor een aantal x waarden tussen min en max.

2.4 De procedure GRAPH.

Om de volledige functie af te beelden gaan we gebruik maken van een procedure graph.

to GRAPH :regel :x :max



local "statelist


make "statelist pos



pu


plot :regel :x :max



pu


setpos :statelist



pd

end
Deze procedure krijgt als argumenten de regel of de functie mee, de minimum waarde voor x die dus eigenlijk de eerste x waarde is de we gaan evalueren en daarom onmiddellijk opslaan in x en de maximum waarde voor x.

Alvorens we kunnen beginnen met tekenen gaan we de beginpositie van de turtle opslaan in statelist zodat we de turtle op het einde kunnen terug zetten naar zijn beginpositie.

Het is de procedure plot die de eigenlijke afbeelding maakt.

2.5 De procedure PLOT.

to PLOT :regel :x :max



if :x > :max [stop]



setpos list :xschaal * :x :yschaal * evalx :regel :x



pd



plot :regel :x + :inc :max

end
Eerst wordt gecontroleerd of de waarde x het maximum nog niet overstegen heeft, als dit zo is mogen we stoppen.

Anders gaan we de eerste x evalueren en de turtle naar het punt verzetten, zonder iets te tekenen want voor de procedure werd opgeroepen in graph hebben we penup gedaan.

Eens de turtle op het eerst punt staat mogen we beginnen tekenen : pendown.

Nu gaan we een recursieve oproep naar plot doen met de zelfde regel en de zelfde waarde voor max maar met een andere waarde voor x. We gaan de waarde van x verhogen met inc.

Het instellen van de waarde van inc gebeurd met de procedure:


to SETINC :num



make "inc :num

end
We maken van het instellen van de waarde van inc een procedure en zetten die waarde niet rechtstreeks in de procedure plot , zo kunnen we heel gemakkelijk de waarde ervan veranderen.

2.6 Voorbeelden.

Maak eerst het scherm proper.

1. teken een assenstelsel met schaalfactoren 10 10

2. Zet de waarde van inc op 1.

3. teken de functie “y = 2 x + 5” voor x gaande van –10 tot 10

4. Teken de functie “ y = (x-3) (x+2)” Zoek zelf uit wat de beste grenzen zijn.

5. Teken een nieuw assenstelsel met schaalfactoren 50 50

6. Teken nu de functie “x * x”van –3 tot 3

Je ziet hoeken in de functie ipv mooie rondingen.

Dit komt omdat de waarde van inc veel te groot is.

7. Teken een assenstelsel met schaalfactoren 20 voor x en 150 voor y en set de waarde van inc op 0.1

8. Teken nu de functies van radsin x en radcos x

Waar je dus moet opletten als je functies wil plotten is dat je de juiste grenzen kiest voor x. Dat de waarde van f(x) niet buiten het assenstelsel valt. En dat de waarde van inc niet te groot is, want dan krijg je een verkeerd beeld. Te klein is niet zo erg, dan teken je gewoon te veel punten.

HOOFDSTUK 3 : Het gebruik van functies.

3.1 Waarvoor kunnen we het plotten van functies gebruiken.

Het plotten van functies kan gebruikt worden in de lessen wiskunde of fysica of…

Door een functie te plotten ( je kan dit doen in LOGO maar ook op een blad papier) kan je veel vlugger een schatting maken van de snijpunten van een functie met de assen. Of kan je de snijpunten van 2 functies vinden, dit kan je dan gebruiken voor het oplossen van stelsel en dergelijke.

Het kan ook gebruikt worden om de minima en maxima te vinden van een functie, dit kan je gebruiken bij afgeleiden en integralen.

Er zijn natuurlijk ook andere toepassingen, je moet zelf maar eens denken waarvoor je zoiets nog kan gebruiken in de andere lessen.

3.2 Voorbeelden.

Zoek voor volgende voorbeelden zelf uit wat de beste waarden zijn voor inc en voor de grenzen.

1. Bepaal ( ongeveer) de snijpunten van de assen met de volgende functie : y = 0.2 x + 2

2. Zoek de snijpunten (ongeveer) van de volgende 2” functies :

Y = 0.(x-5)(x+4) en y = 0.5(x+5)







PAGE  
11

_1036835355

